# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Philipp Schmidt <philschmidt at gmx dot net>
# Contributor: Stefan Tatschner <stefan.tatschner at gmail dot com>

pkgname=python-blinker
pkgver=1.5
pkgrel=2
pkgdesc="Fast, simple object-to-object and broadcast signaling"
arch=('any')
url="https://blinker.readthedocs.io"
license=('MIT')
depends=('python')
makedepends=('python-setuptools')
checkdepends=('python-pytest')
source=("https://files.pythonhosted.org/packages/source/b/blinker/blinker-$pkgver.tar.gz")
sha512sums=('b1880fdc974be63e16e2b826bdbc8ee161bc0234591b8b41deed937a1e1b9d0bf1fb697c8e94644de9cc1900aedd39d08cfc3e59fef51abec295552f96418722')

check() {
  cd blinker-$pkgver
  python -m pytest
}

package_python-blinker() {
  cd blinker-$pkgver
  python setup.py install --root="$pkgdir/" --optimize=1
  install -Dm644 LICENSE.rst -t "$pkgdir"/usr/share/licenses/$pkgname/
}
